Series de Fourier de funciones pares o impares.

Onda cuadrada 1 (Período T, intervalo (-T/2,T/2))

```
> restart:
> with(plots): setoptions(thickness=1):
> T:=2: Digits:=7:
```

Consideremos la siguiente función onda cuadrada, de período T, en el intervalo [-T/2, t/2]

```
> f:=piecewise((-T/2<=t and t<0,-1),(0<=t and t<=T/2,1));
  #latex(%);
```

$$f := \begin{cases} -1 & -1 \leq t \textbf{ and } t < 0 \\ 1 & 0 \leq t \textbf{ and } t \leq 1 \end{cases}$$

```
> plot(f,t=-T/2..T/2);
```




```
> N:=20:t0:=-T/2: t1:=T/2:
```

Calculemos los coeficientes de Fourier y los coeficientes del espcctro de potencia

```
> for n from 0 to N do

  a[n]:=2/T*int(f*cos(n*2*Pi/T*t),t=t0..t1):
```

```
b[n]:=2/T*int(f*sin(n*2*Pi/T*t),t=t0..t1):

A[n]:=sqrt(a[n]^2+b[n]^2):

phi[n]:=argument((b[n]+1E-10)+I*a[n]):
od:
```

Con ellos construimos la serie de Fourier

```
> SerieFourier := (m,t)->
                    a[0]/2 +
                    sum(a[k]*cos((2*k*Pi*t)/T),k=1..m) +
          sum(b[k]*sin((2*k*Pi*t)/T),k=1..m) ;
```

$$SerieFourier := (m, t) \rightarrow \frac{1}{2}\, a_0 + \sum_{k=1}^{m} a_k \cos\left(\frac{2\, k\, \pi\, t}{T}\right) + \sum_{k=1}^{m} b_k \sin\left(\frac{2\, k\, \pi\, t}{T}\right)$$

Verificamos algunas expansiones para n=5 y n=10

```
> SerieFourier(5,t);SerieFourier(10,t);
```

$$\frac{4 \sin(\pi\, t)}{\pi} + \frac{4}{3}\, \frac{\sin(3\, \pi\, t)}{\pi} + \frac{4}{5}\, \frac{\sin(5\, \pi\, t)}{\pi}$$

$$\frac{4 \sin(\pi\, t)}{\pi} + \frac{4}{3}\, \frac{\sin(3\, \pi\, t)}{\pi} + \frac{4}{5}\, \frac{\sin(5\, \pi\, t)}{\pi} + \frac{4}{7}\, \frac{\sin(7\, \pi\, t)}{\pi} + \frac{4}{9}\, \frac{\sin(9\, \pi\, t)}{\pi}$$

```
> plot([f,SerieFourier(3,t),SerieFourier(7,t), SerieFourier(20,t)
  ],t=-T..T,title="Aproximaciones de Fourier: n=3 (Verde),n=7
  (Azul), n=20 (negro) ",color=[red,green,blue,black],numpoints=
  100);
  plot([f,SerieFourier(3,t),SerieFourier(7,t), SerieFourier(20,t)
  ],t=-2*T..2*T,title="Aproximaciones de Fourier: n=3 (Verde),n=7
  (Azul), n=20 (negro) ",color=[red,green,blue,black],numpoints=
  100);
```


Aproximaciones de Fourier: n=3 (Verde),n=7 (Azul), n=20 (negro)
1
0.5
−0.5
−1
−2
−1
0
1
2
t




```
> SeProm:=evalf(a[0]/2):
```

El espectro de potencia se puede graficar como

```
> Amp_a0:=plot([[0,0],[0,SeProm]],thickness=3 ):
  Amp_coef:=[seq(plot([[n,0],[n,A[n]]]),n=1..N)]:
  display(Amp_a0,Amp_coef,title=`Espectro de Potencia de la
  Señal`);
```



Onda cuadrada 2

¿ qué hubiera pasado si el intervalo de integración, o el período hubiera sido diferente ?
Obvio que es la misma función pero la contruimos de manera distinta. Consideremos la misma función sólo que diferente:

```
> restart:
> with(plots):
> T:=2: Digits:=7:
```

Consideremos la siguiente función onda cuadrada, de período T, en el intervalo [-T/2, t/2]

```
> f:=piecewise((0 <=t and t<T/2,1),(T/2<=t and t<=T,-1));
  #latex(%);
```

$$f := \begin{cases} 1 & 0 \le t \textbf{ and } t < 1 \\ -1 & 1 \le t \textbf{ and } t \le 2 \end{cases}$$

```
> plot(f,t=0..T);
```




```
> N:=20:t0:=0: t1:=T:
```

Calculemos los coeficientes de Fourier y los coeficientes espectrales

```
> for n from 0 to N do

  a[n]:=2/T*int(f*cos(n*2*Pi/T*t),t=t0..t1):

  b[n]:=2/T*int(f*sin(n*2*Pi/T*t),t=t0..t1):

  A[n]:=sqrt(a[n]^2 + b[n]^2):
  phi[n]:=argument((b[n]+1E-10)+I*a[n]):
  od:
```

Con ellos construimos la serie de Fourier

```
> SerieFourier := (m,t)->
                        a[0]/2 +
                        sum(a[k]*cos((2*k*Pi*t)/T),k=1..m) +
              sum(b[k]*sin((2*k*Pi*t)/T),k=1..m) ;
```

$$SerieFourier := (m, t) \rightarrow \frac{1}{2}\, a_0 + \sum_{k=1}^{m} a_k \cos\left(\frac{2\, k\, \pi\, t}{T}\right) + \sum_{k=1}^{m} b_k \sin\left(\frac{2\, k\, \pi\, t}{T}\right)$$

Verificamos algunas expansiones para n=5 y n=10

```
> SerieFourier(5,t);SerieFourier(10,t);
```

$$\frac{4\sin(\pi t)}{\pi} + \frac{4}{3}\frac{\sin(3\pi t)}{\pi} + \frac{4}{5}\frac{\sin(5\pi t)}{\pi}$$

$$\frac{4\sin(\pi t)}{\pi} + \frac{4}{3}\frac{\sin(3\pi t)}{\pi} + \frac{4}{5}\frac{\sin(5\pi t)}{\pi} + \frac{4}{7}\frac{\sin(7\pi t)}{\pi} + \frac{4}{9}\frac{\sin(9\pi t)}{\pi}$$

Como es la misma función expresada en la base de Fourier, obviamente dan los mismos coeficientes. Con ello, la conclusión es que uno puede escoger a voluntad el intervalo (si es la misma función) para que las integrales sean más fáciles de evaluar.

claramente las gráficas serán las mismas.

```
> plot([f,SerieFourier(3,t),SerieFourier(7,t), SerieFourier(20,t)
  ],t=-T..T,title="Aproximaciones de Fourier: n=3 (Verde),n=7
  (Azul), n=20 (negro) ",color=[red,green,blue,black],numpoints=
  100);

  plot([f,SerieFourier(3,t),SerieFourier(7,t), SerieFourier(20,t)
  ],t=-3*T..3*T,title="Aproximaciones de Fourier: n=3 (Verde),n=7
  (Azul), n=20 (negro) ",color=[red,green,blue,black],numpoints=
  100);
```

Aproximaciones de Fourier: n=3 (Verde),n=7 (Azul), n=20 (negro)





y el espectro de potencia, también será el mismo....

```
> SeProm:=evalf(a[0]/2):
> Amp_a0:=plot([[0,0],[0,SeProm]],thickness=3):

  Amp_coef:=[seq(plot([[n,0],[n,A[n]]]),n=1..N)]:
  display(Amp_a0,Amp_coef,title=`Espectro de la Señal`);
```



Diente de sierra 1 Intervalo [0,T]

Consideremos ahora la funci'on diente de sierra

```
> restart:
> with(plots): setoptions(thickness=2): assume(k,integer):
> T:=2: Digits:=7:t0:=0: t1:=T:
```

Esta funci'on viene descrita como

```
> f:=piecewise((t0<=t and t<t1,3*t));
```

$$f := \begin{cases} 3\,t & 0 \le t \textbf{ and } t < 2 \\ 0 & \textit{otherwise} \end{cases}$$

```
> plot(f,t=t0..t1);
```



calculamos entonces los 20 primeros términos de la Serie de Fourier para esta función.

```
> N:=20:
> for n from 0 to N do

  a[n]:=2/T*int(f*cos(n*2*Pi/T*t),t=t0..t1):
  b[n]:=2/T*int(f*sin(n*2*Pi/T*t),t=t0..t1):
  A[n]:=sqrt(a[n]^2+b[n]^2):
  phi[n]:=argument((b[n]+1E-10)+I*a[n]):
  od:
```

anal'iticamente hubiera sido

```
> aa[0]:=2/TT*int(a*x,x=0..TT);
```

$$aa_0 := TT\, a$$

para el amónico fundamental

```
> aa[k]:=2/TT*int(a*x*cos(k*2*Pi/TT*x),x=0..TT);
```

$$aa_{k\sim} := 0$$

para los armónicos pares de orden superior y

```
> bb[k]:=2/TT*int(a*x*sin(k*2*Pi/TT*x),x=0..TT);#latex(%);
```

$$bb_{k\sim} := -\frac{TT\, a}{k\sim\, \pi}$$

para la contribución de los armónicos impares

```
> a[0],a[4],a[7],b[3],b[9];
```

$$6, 0, 0, -\frac{2}{\pi}, -\frac{2}{3\pi}$$

Construimos, entonces la serie de Fourier

```
> SerieFourier := (m,t)->
                    a[0]/2 +
                    sum(a[k]*cos((2*k*Pi*t)/T),k=1..m) +
             sum(b[k]*sin((2*k*Pi*t)/T),k=1..m) ;
```

$$SerieFourier := (m, t) \to \frac{1}{2} a_0 + \sum_{k=1}^{m} a_k \cos\left(\frac{2 k \pi t}{T}\right) + \sum_{k=1}^{m} b_k \sin\left(\frac{2 k \pi t}{T}\right)$$

y evaluamos la serie para un par de desarrollos posibles n = 5 y n=10

```
> SerieFourier(5,t);#latex(%);
  SerieFourier(10,t);
```

$$3 - \frac{6\sin(\pi t)}{\pi} - \frac{3\sin(2\pi t)}{\pi} - \frac{2\sin(3\pi t)}{\pi} - \frac{3}{2}\frac{\sin(4\pi t)}{\pi} - \frac{6}{5}\frac{\sin(5\pi t)}{\pi}$$

$$3 - \frac{6\sin(\pi t)}{\pi} - \frac{3\sin(2\pi t)}{\pi} - \frac{2\sin(3\pi t)}{\pi} - \frac{3}{2}\frac{\sin(4\pi t)}{\pi} - \frac{6}{5}\frac{\sin(5\pi t)}{\pi} - \frac{\sin(6\pi t)}{\pi} - \frac{6}{7}\frac{\sin(7\pi t)}{\pi} - \frac{3}{4}\frac{\sin(8\pi t)}{\pi} - \frac{2}{3}\frac{\sin(9\pi t)}{\pi} - \frac{3}{5}\frac{\sin(10\pi t)}{\pi}$$

graficamos las representaciones de la función para n=3, n=7 n=10

```
> plot([f,SerieFourier(3,t),SerieFourier(7,t), SerieFourier(10,t)
  ],t=-T..T,title="Aproximaciones de Fourier: n=3 (Verde),n=7
  (Azul), n=10 (negro) ",color=[red,green,blue,black],numpoints=
  100);
```



El promedio de la función será la contribución del armónico fundamental

```
> SeProm:=evalf(a[0]/2):
```

y esta será la contribución del resto de los armónicos al espectro de potencia

```
> Amp_a0:=plot([[0,0],[0,SeProm]],thickness=3 ):
  Amp_coef:=[seq(plot([[n,0],[n,A[n]]]),n=1..N)]:
  display(Amp_a0,Amp_coef,title=`Espectro de la Señal`);
```



Diente de sierra 2 (impar) Intervalo [-T/2,T/2]

```
> restart:
> with(plots): setoptions(thickness=2): assume(k,integer):
> T:=2:Digits:=7:t0:=-T/2: t1:=T/2:
> f:=piecewise((t0<=t and t<=t1,3*t));
```

$$f := \begin{cases} 3\,t & -1 \leq t \text{ and } t \leq 1 \\ 0 & \textit{otherwise} \end{cases}$$

```
> plot(f,t=t0..t1);
```




```
> N:=20:
> for i from 0 to N do

  a[i]:=2/T*int(f*cos(i*2*Pi/T*t),t=t0..t1):
  b[i]:=2/T*int(f*sin(i*2*Pi/T*t),t=t0..t1):
  A[i]:=sqrt(a[i]^2+b[i]^2):
  phi[i]:=argument((b[i]+1E-10)+I*a[i]):
  od:
```

anal'iticamente hubiera sido

```
> aa[0]:=2/TT*int(a*x,x=-TT/2..TT/2);
```

$$aa_0 := 0$$

para el amónico fundamental

```
> aa[k]:=2/TT*int(a*x*cos(k*2*Pi/TT*x),x=-TT/2..TT/2);
```

$$aa_{k\sim} := 0$$

para los armónicos pares de orden superior y

```
> bb[k]:=2/TT*int(a*x*sin(k*2*Pi/TT*x),x=-TT/2..TT/2);#latex(%);
```

$$bb_{k\sim} := -\frac{TT\, a\, (-1)^{k\sim}}{k\sim\, \pi}$$

```
> a[0],a[4],a[7],b[3],b[9];
```

$$0,\ 0,\ 0,\ \frac{2}{\pi},\ \frac{2}{3\,\pi}$$

```
> SerieFourier := (m,t)->
                    a[0]/2 +
                    sum(a[k]*cos((2*k*Pi*t)/T),k=1..m) +
           sum(b[k]*sin((2*k*Pi*t)/T),k=1..m) ;
```

$$SerieFourier := (m, t) \to \frac{1}{2}\,a_0 + \sum_{k=1}^{m} a_k \cos\left(\frac{2\,k\,\pi\,t}{T}\right) + \sum_{k=1}^{m} b_k \sin\left(\frac{2\,k\,\pi\,t}{T}\right)$$

```
> SerieFourier(5,t); # latex(%);
```

$$\frac{6\sin(\pi\,t)}{\pi} - \frac{3\sin(2\,\pi\,t)}{\pi} + \frac{2\sin(3\,\pi\,t)}{\pi} - \frac{3}{2}\,\frac{\sin(4\,\pi\,t)}{\pi} + \frac{6}{5}\,\frac{\sin(5\,\pi\,t)}{\pi}$$

```
> SerieFourier(10,t);
```

$$\frac{6\sin(\pi\,t)}{\pi} - \frac{3\sin(2\,\pi\,t)}{\pi} + \frac{2\sin(3\,\pi\,t)}{\pi} - \frac{3}{2}\,\frac{\sin(4\,\pi\,t)}{\pi} + \frac{6}{5}\,\frac{\sin(5\,\pi\,t)}{\pi}$$
$$- \frac{\sin(6\,\pi\,t)}{\pi} + \frac{6}{7}\,\frac{\sin(7\,\pi\,t)}{\pi} - \frac{3}{4}\,\frac{\sin(8\,\pi\,t)}{\pi} + \frac{2}{3}\,\frac{\sin(9\,\pi\,t)}{\pi}$$
$$- \frac{3}{5}\,\frac{\sin(10\,\pi\,t)}{\pi}$$

```
> plot([f,SerieFourier(3,t),SerieFourier(7,t), SerieFourier(20,t)
  ],t=-T..T,title="Aproximaciones de Fourier: n=3 (Verde),n=7
  (Azul), n=20 (negro) ",color=[red,green,blue,black],numpoints=
  100);
```




```
> SeProm:=evalf(a[0]/2):
> Amp_a0:=plot([[0,0],[0,SeProm]],thickness=3 ):
                Amp_coef:=[seq(plot([[n,0],[n,A[n]]]),n=1..N)]:
                              display(Amp_a0,Amp_coef,title=
  `Espectro de la Señal`);
```



Diente de Sierra 3 (par) Intervalo [-T/2,T/2]

```
> restart:
> with(plots):  setoptions(thickness=2):
> T:=2: Digits:=7:t0:=-T/2: t1:=T/2:
> f:=piecewise((t0<=t and t<=0,-3*t),(0<t and t<=t1,3*t));
```

$$f := \begin{cases} -3\,t & -1 \le t \text{ and } t \le 0 \\ 3\,t & 0 < t \text{ and } t \le 1 \end{cases}$$

```
> plot(f,t=t0..t1);
```




```
> N:=20:
> for n from 0 to N do

  a[n]:=2/T*int(f*cos(n*2*Pi/T*t),t=t0..t1):
  b[n]:=2/T*int(f*sin(n*2*Pi/T*t),t=t0..t1):
  A[n]:=sqrt(a[n]^2+b[n]^2):
  phi[n]:=argument((b[n]+1E-10)+I*a[n]):
  od:
```

anal'iticamente hubiera sido

```
> assume(k,integer):
> aa[0]:=2/TT*(int(-a*x,x=-TT/2..0) + int(a*x,x=0..TT/2));
```

$$aa_0 := \frac{1}{2}\,TT\,a$$

para el amónico fundamental

```
> aa[k]:=2/TT*(int(-a*x*cos(k*2*Pi/TT*x),x=-TT/2..0) +
                int(a*x*cos(k*2*Pi/TT*x),x=0..TT/2));
Error, (in IntegrationTools:-Definite:-Main) too many levels of 
recursion
```

para los armónicos pares de orden superior y

```
> bb[k]:=2/TT*(int(-a*x*sin(k*2*Pi/TT*x),x=-TT/2..0) +
               int(a*x*sin(k*2*Pi/TT*x),x=0..TT/2));#latex(%);
```

Error, (in IntegrationTools:-Definite:-Main) too many levels of recursion

```
> a[0],a[4],a[7],b[3],b[9];
```

$$3, 0, -\frac{12}{49\,\pi^2}, 0, 0$$

```
> SerieFourier := (m,t)->
                    a[0]/2 +
                    sum(a[k]*cos((2*k*Pi*t)/T),k=1..m) +
              sum(b[k]*sin((2*k*Pi*t)/T),k=1..m) ;
```

$$SerieFourier := (m, t) \to \frac{1}{2}\,a_0 + \sum_{k=1}^{m} a_k \cos\left(\frac{2\,k\,\pi\,t}{T}\right) + \sum_{k=1}^{m} b_k \sin\left(\frac{2\,k\,\pi\,t}{T}\right)$$

```
> SerieFourier(5,t);#latex(%);
```

$$\frac{3}{2} - \frac{12\cos(\pi\,t)}{\pi^2} - \frac{4}{3}\,\frac{\cos(3\,\pi\,t)}{\pi^2} - \frac{12}{25}\,\frac{\cos(5\,\pi\,t)}{\pi^2}$$

```
> SerieFourier(10,t);
```

$$\frac{3}{2} - \frac{12\cos(\pi\,t)}{\pi^2} - \frac{4}{3}\,\frac{\cos(3\,\pi\,t)}{\pi^2} - \frac{12}{25}\,\frac{\cos(5\,\pi\,t)}{\pi^2} - \frac{12}{49}\,\frac{\cos(7\,\pi\,t)}{\pi^2}$$

$$- \frac{4}{27}\,\frac{\cos(9\,\pi\,t)}{\pi^2}$$

```
> plot([f,SerieFourier(3,t),SerieFourier(7,t), SerieFourier(20,t)
  ],t=-T..T,title="Aproximaciones de Fourier: n=3 (Verde),n=7
  (Azul), n=20 (negro) ",color=[red,green,blue,black],numpoints=
  100);
```




```
> SeProm:=evalf(a[0]/2):
> Amp_a0:=plot([[0,0],[0,SeProm]],thickness=3 ):
  Amp_coef:=[seq(plot([[n,0],[n,A[n]]]),n=1..N)]:
  display(Amp_a0,Amp_coef,title=`Espectro de la Señal`);
```



Parábola invertida

```
> restart:
> with(plots):
> T:=2: Digits:=7:t0:=-T/2: t1:=T/2:
> f:=piecewise((t0<=t and t<=t1,t^2));
```

$$f := \begin{cases} t^2 & -1 \le t \text{ and } t \le 1 \\ 0 & \text{otherwise} \end{cases}$$

```
> plot(f,t=t0..t1);
```




```
> N:=20:
> for n from 0 to N do

  a[n]:=2/T*int(f*cos(n*2*Pi/T*t),t=t0..t1):
  b[n]:=2/T*int(f*sin(n*2*Pi/T*t),t=t0..t1):

  A[n]:=sqrt(a[n]^2+b[n]^2):
  phi[n]:=argument((b[n]+1E-10)+I*a[n]):
  od:
> SerieFourier := (m,t)->
                          a[0]/2 +
                          sum(a[k]*cos((2*k*Pi*t)/T),k=1..m) +
              sum(b[k]*sin((2*k*Pi*t)/T),k=1..m) ;
```

$$SerieFourier := (m, t) \to \frac{1}{2}\, a_0 + \sum_{k=1}^{m} a_k \cos\left(\frac{2\, k\, \pi\, t}{T}\right) + \sum_{k=1}^{m} b_k \sin\left(\frac{2\, k\, \pi\, t}{T}\right)$$

```
> SerieFourier(5,t);
```

$$\frac{1}{3} - \frac{4 \cos(\pi\, t)}{\pi^2} + \frac{\cos(2\, \pi\, t)}{\pi^2} - \frac{4}{9}\, \frac{\cos(3\, \pi\, t)}{\pi^2} + \frac{1}{4}\, \frac{\cos(4\, \pi\, t)}{\pi^2}$$

$$- \frac{4}{25}\, \frac{\cos(5\, \pi\, t)}{\pi^2}$$

```
> SerieFourier(10,t);
```

$$\frac{1}{3} - \frac{4\cos(\pi t)}{\pi^2} + \frac{\cos(2\pi t)}{\pi^2} - \frac{4}{9}\frac{\cos(3\pi t)}{\pi^2} + \frac{1}{4}\frac{\cos(4\pi t)}{\pi^2} - \frac{4}{25}\frac{\cos(5\pi t)}{\pi^2} + \frac{1}{9}\frac{\cos(6\pi t)}{\pi^2} - \frac{4}{49}\frac{\cos(7\pi t)}{\pi^2} + \frac{1}{16}\frac{\cos(8\pi t)}{\pi^2} - \frac{4}{81}\frac{\cos(9\pi t)}{\pi^2} + \frac{1}{25}\frac{\cos(10\pi t)}{\pi^2}$$

```
> plot([f,SerieFourier(3,t),SerieFourier(5,t), SerieFourier(10,t)
  ],t=-T..T,title="Aproximaciones de Fourier: n=3 (Verde),n=7
  (Azul), n=20 (negro) ",color=[red,green,blue,black],numpoints=
  100);
  plot([f,SerieFourier(3,t),SerieFourier(5,t), SerieFourier(10,t)
  ],t=-3*T..3*T,title="Aproximaciones de Fourier: n=3 (Verde),n=7
  (Azul), n=20 (negro) ",color=[red,green,blue,black],numpoints=
  100);
```






```
> SeProm:=evalf(a[0]/2):
> Amp_a0:=plot([[0,0],[0,SeProm]],thickness=3 ):
                 Amp_coef:=[seq(plot([[n,0],[n,A[n]]]),n=1..N)]:
                                  display(Amp_a0,Amp_coef,title=
  `Espectro de la Señal`);
```



Una cuerda de longitud L

Sobre el eje x, consideremos una cuerda de longitud L fija en sus dos extremos. En x = xL/4 se desplaza y0. Encentre las expansiones en series de Fourier

```
> restart:
> with(plots):
> Digits:=7: L :=2: tp := L/4; y0:=L/10;
```

$$tp := \frac{1}{2}$$

$$y0 := \frac{1}{5}$$

```
> f:=piecewise(( 0<=t and t<=tp,4*y0*t/L),
  (tp<t and t<=L,(4*y0/3)*(-t/L +1)) );
```

$$f := \begin{cases} \frac{2}{5}\,t & 0 \le t \text{ and } t \le \frac{1}{2} \\ -\frac{2}{15}\,t + \frac{4}{15} & \frac{1}{2} < t \text{ and } t \le 2 \end{cases}$$

```
> plot(f,t=0..L);
```



La serie con un período L

```
> T:=L: t0:=0: t1:=L:
> N:=20:
> for n from 0 to N do
      a[n]:=2/T*int(f*cos(n*2*Pi/T*t),t=t0..t1):
      b[n]:=2/T*int(f*sin(n*2*Pi/T*t),t=t0..t1):
      A[n]:=sqrt(a[n]^2+b[n]^2):
      phi[n]:=argument((b[n]+1E-10)+I*a[n]):
  od:
> a[0],a[4],a[7],b[3],b[9];
```

$$\frac{1}{5},\, 0,\, -\frac{1}{245}\,\frac{2+7\,\pi}{\pi^2} + \frac{1}{735}\,\frac{21\,\pi - 2}{\pi^2},\, -\frac{8}{135\,\pi^2},\, \frac{8}{1215\,\pi^2}$$

```
> SerieFourier := (m,t)->
```

```
        a[0]/2 +
        sum(a[k]*cos((2*k*Pi*t)/T),k=1..m) +
    sum(b[k]*sin((2*k*Pi*t)/T),k=1..m) ;
```

$$SerieFourier := (m,t) \rightarrow \frac{1}{2}a_0 + \sum_{k=1}^{m} a_k \cos\left(\frac{2k\pi t}{T}\right) + \sum_{k=1}^{m} b_k \sin\left(\frac{2k\pi t}{T}\right)$$

```
> SerieFourier(5,t);simplify(%);
```

$$\frac{1}{10} + \left(\frac{1}{5}\frac{-2+\pi}{\pi^2} - \frac{1}{15}\frac{3\pi+2}{\pi^2}\right)\cos(\pi t) - \frac{4}{15}\frac{\cos(2\pi t)}{\pi^2} + \left(-\frac{1}{45}\frac{3\pi+2}{\pi^2} + \frac{1}{135}\frac{9\pi-2}{\pi^2}\right)\cos(3\pi t) + \left(\frac{1}{125}\frac{-2+5\pi}{\pi^2} - \frac{1}{375}\frac{15\pi+2}{\pi^2}\right)\cos(5\pi t) + \frac{8}{15}\frac{\sin(\pi t)}{\pi^2} - \frac{8}{135}\frac{\sin(3\pi t)}{\pi^2} + \frac{8}{375}\frac{\sin(5\pi t)}{\pi^2}$$

$$-\frac{1}{6750}\frac{1}{\pi^2}(-675\pi^2 + 3600\cos(\pi t) + 1800\cos(2\pi t) + 400\cos(3\pi t) + 144\cos(5\pi t) - 3600\sin(\pi t) + 400\sin(3\pi t) - 144\sin(5\pi t))$$

```
> SerieFourier(10,t);simplify(%);
```

$$\frac{1}{10} + \left(\frac{1}{5}\frac{-2+\pi}{\pi^2} - \frac{1}{15}\frac{3\pi+2}{\pi^2}\right)\cos(\pi t) - \frac{4}{15}\frac{\cos(2\pi t)}{\pi^2} + \left(-\frac{1}{45}\frac{3\pi+2}{\pi^2} + \frac{1}{135}\frac{9\pi-2}{\pi^2}\right)\cos(3\pi t) + \left(\frac{1}{125}\frac{-2+5\pi}{\pi^2} - \frac{1}{375}\frac{15\pi+2}{\pi^2}\right)\cos(5\pi t) - \frac{4}{135}\frac{\cos(6\pi t)}{\pi^2} + \left(-\frac{1}{245}\frac{2+7\pi}{\pi^2} + \frac{1}{735}\frac{21\pi-2}{\pi^2}\right)\cos(7\pi t) + \left(\frac{1}{405}\frac{9\pi-2}{\pi^2} - \frac{1}{1215}\frac{27\pi+2}{\pi^2}\right)\cos(9\pi t) - \frac{4}{375}\frac{\cos(10\pi t)}{\pi^2} + \frac{8}{15}\frac{\sin(\pi t)}{\pi^2} - \frac{8}{135}\frac{\sin(3\pi t)}{\pi^2} + \frac{8}{375}\frac{\sin(5\pi t)}{\pi^2} - \frac{8}{735}\frac{\sin(7\pi t)}{\pi^2} + \frac{8}{1215}\frac{\sin(9\pi t)}{\pi^2}$$

$$-\frac{1}{2976750}\frac{1}{\pi^2}(-297675\pi^2 + 1587600\cos(\pi t) + 793800\cos(2\pi t) + 176400\cos(3\pi t) + 63504\cos(5\pi t) + 88200\cos(6\pi t) + 32400\cos(7\pi t)$$

$+ 19600 \cos(9 \pi t) + 31752 \cos(10 \pi t) - 1587600 \sin(\pi t)$

$+ 176400 \sin(3 \pi t) - 63504 \sin(5 \pi t) + 32400 \sin(7 \pi t) - 19600 \sin(9 \pi t))$

```
> plot([f,SerieFourier(3,t),SerieFourier(7,t), SerieFourier(10,
t)],t=-T..T,title="Aproximaciones de Fourier: n=3 (Verde),n=7
(Azul), n=10 (negro) ",color=[red,green,blue,black],
numpoints=100);
```




```
> SeProm:=evalf(a[0]/2):
> Amp_a0:=plot([[0,0],[0,SeProm]],thickness=3 ):
Amp_coef:=[seq(plot([[n,0],[n,A[n]]]),n=1..N)]:
display(Amp_a0,Amp_coef,title=`Espectro de la Señal`);
```



La Serie antisimétrica respecto al eje x = 0 (período 2 L∯

```
> f:=piecewise( (-L <=t and t<-tp,(4*y0/3)*(-t/L -1)),
                   ( -tp<=t and t<=tp,4*y0*t/L),
          (tp<t and t<=L,(4*y0/3)*(-t/L +1)) );
```

$$f := \begin{cases} -\frac{2}{15}t - \frac{4}{15} & -2 \leq t \text{ and } t < -\frac{1}{2} \\ \frac{2}{5}t & -\frac{1}{2} \leq t \text{ and } t \leq \frac{1}{2} \\ -\frac{2}{15}t + \frac{4}{15} & \frac{1}{2} < t \text{ and } t \leq 2 \end{cases}$$

```
> plot(f,t=-L..L);
```




```
> T:=2*L: t0:=-L: t1:=L:
> N:=20:
> for n from 0 to N do

  a[n]:=2/T*int(f*cos(n*2*Pi/T*t),t=t0..t1):
  b[n]:=2/T*int(f*sin(n*2*Pi/T*t),t=t0..t1):
  A[n]:=sqrt(a[n]^2+b[n]^2):
          phi[n]:=argument((b[n]+1E-10)+I*a[n]):
  od:
> a[0],a[4],a[7],b[3],b[9];
```

$$0, 0, 0, -\frac{1}{135}\,\frac{\sqrt{2}\,(-4+9\,\pi)}{\pi^2}+\frac{1}{45}\,\frac{\sqrt{2}\,(4+3\,\pi)}{\pi^2}, \frac{1}{1215}\,\frac{\sqrt{2}\,(4+27\,\pi)}{\pi^2} - \frac{1}{405}\,\frac{\sqrt{2}\,(-4+9\,\pi)}{\pi^2}$$

```
> SerieFourier := (m,t)->
                    a[0]/2 +
                    sum(a[k]*cos((2*k*Pi*t)/T),k=1..m) +
          sum(b[k]*sin((2*k*Pi*t)/T),k=1..m) ;
```

$$SerieFourier := (m,t) \to \frac{1}{2} a_0 + \sum_{k=1}^{m} a_k \cos\left(\frac{2 k \pi t}{T}\right) + \sum_{k=1}^{m} b_k \sin\left(\frac{2 k \pi t}{T}\right)$$

```
> SerieFourier(5,t);simplify(%);
```

$$\left(\frac{1}{15} \frac{\sqrt{2}\,(4+3\pi)}{\pi^2} - \frac{1}{5} \frac{\sqrt{2}\,(-4+\pi)}{\pi^2}\right) \sin\left(\frac{1}{2}\pi t\right) + \frac{8}{15} \frac{\sin(\pi t)}{\pi^2} + \left(-\frac{1}{135} \frac{\sqrt{2}\,(-4+9\pi)}{\pi^2} + \frac{1}{45} \frac{\sqrt{2}\,(4+3\pi)}{\pi^2}\right) \sin\left(\frac{3}{2}\pi t\right) + \left(-\frac{1}{375} \frac{\sqrt{2}\,(4+15\pi)}{\pi^2} + \frac{1}{125} \frac{\sqrt{2}\,(-4+5\pi)}{\pi^2}\right) \sin\left(\frac{5}{2}\pi t\right)$$

$$\frac{8}{3375} \frac{1}{\pi^2} \left(450\sqrt{2} \sin\left(\frac{1}{2}\pi t\right) + 225 \sin(\pi t) + 50\sqrt{2} \sin\left(\frac{3}{2}\pi t\right) - 18\sqrt{2} \sin\left(\frac{5}{2}\pi t\right)\right)$$

```
> SerieFourier(10,t);simplify(%);
```

$$\left(\frac{1}{15} \frac{\sqrt{2}\,(4+3\pi)}{\pi^2} - \frac{1}{5} \frac{\sqrt{2}\,(-4+\pi)}{\pi^2}\right) \sin\left(\frac{1}{2}\pi t\right) + \frac{8}{15} \frac{\sin(\pi t)}{\pi^2} + \left(-\frac{1}{135} \frac{\sqrt{2}\,(-4+9\pi)}{\pi^2} + \frac{1}{45} \frac{\sqrt{2}\,(4+3\pi)}{\pi^2}\right) \sin\left(\frac{3}{2}\pi t\right) + \left(-\frac{1}{375} \frac{\sqrt{2}\,(4+15\pi)}{\pi^2} + \frac{1}{125} \frac{\sqrt{2}\,(-4+5\pi)}{\pi^2}\right) \sin\left(\frac{5}{2}\pi t\right) - \frac{8}{135} \frac{\sin(3\pi t)}{\pi^2} + \left(\frac{1}{735} \frac{\sqrt{2}\,(-4+21\pi)}{\pi^2} - \frac{1}{245} \frac{\sqrt{2}\,(4+7\pi)}{\pi^2}\right) \sin\left(\frac{7}{2}\pi t\right) + \left(\frac{1}{1215} \frac{\sqrt{2}\,(4+27\pi)}{\pi^2} - \frac{1}{405} \frac{\sqrt{2}\,(-4+9\pi)}{\pi^2}\right) \sin\left(\frac{9}{2}\pi t\right) + \frac{8}{375} \frac{\sin(5\pi t)}{\pi^2}$$

$$-\frac{8}{1488375} \frac{1}{\pi^2} \left(-198450\sqrt{2} \sin\left(\frac{1}{2}\pi t\right) - 99225 \sin(\pi t) - 22050\sqrt{2} \sin\left(\frac{3}{2}\pi t\right) + 7938\sqrt{2} \sin\left(\frac{5}{2}\pi t\right) + 11025 \sin(3\pi t) + 4050\sqrt{2} \sin\left(\frac{7}{2}\pi t\right) - 2450\sqrt{2} \sin\left(\frac{9}{2}\pi t\right) - 3969 \sin(5\pi t)\right)$$

```
> plot([f,SerieFourier(3,t),SerieFourier(7,t), SerieFourier(10,
t)],t=-T..T,y=-0.3..0.3,title="Aproximaciones de Fourier: n=3
(Verde),n=7 (Azul), n=10 (negro) ",color=[red,green,blue,
black],numpoints=100);
```




```
> SeProm:=evalf(a[0]/2):
> Amp_a0:=plot([[0,0],[0,SeProm]],thickness=3):

  Amp_coef:=[seq(plot([[n,0],[n,A[n]]]),n=1..N)]:

  display(Amp_a0,Amp_coef,title=`Espectro de la Señal`);
```



La Serie simétrica respecto al eje x = 0 (período 2 L∯

```
> f:=piecewise(  (-L <=t and t<-tp,(-4*y0/3)*(-t/L -1)),
                                         ( -tp<=t and t<=0,-4*y0*t/L),
                                         ( 0<t and t<=tp,4*y0*t/L),
                                         (tp<t and t<=L,(4*y0/3)*(-t/L
+1)) );
```

$$f := \begin{cases} \frac{2}{15}\,t + \frac{4}{15} & -2 \le t \textbf{ and } t < -\frac{1}{2} \\ -\frac{2}{5}\,t & -\frac{1}{2} \le t \textbf{ and } t \le 0 \\ \frac{2}{5}\,t & 0 < t \textbf{ and } t \le \frac{1}{2} \\ -\frac{2}{15}\,t + \frac{4}{15} & \frac{1}{2} < t \textbf{ and } t \le 2 \end{cases}$$

```
> plot(f,t=-L..L);
```




```
> T:=2*L: t0:=-L: t1:=L:
> N:=20:
> for n from 0 to N do
  a[n]:=2/T*int(f*cos(n*2*Pi/T*t),t=t0..t1):

  b[n]:=2/T*int(f*sin(n*2*Pi/T*t),t=t0..t1):

  A[n]:=sqrt(a[n]^2+b[n]^2):
            phi[n]:=argument((b[n]+1E-10)+I*a[n]):
  od:
> a[0],a[4],a[7],b[3],b[9];
```

$$\frac{1}{5}, -\frac{4}{15\,\pi^2}, \frac{1}{735}\,\frac{8+4\sqrt{2}+21\sqrt{2}\,\pi}{\pi^2} - \frac{1}{245}\,\frac{-4\sqrt{2}+7\sqrt{2}\,\pi+8}{\pi^2}, 0, 0$$

```
> SerieFourier := (m,t)->
                  a[0]/2 +
                  sum(a[k]*cos((2*k*Pi*t)/T),k=1..m) +
          sum(b[k]*sin((2*k*Pi*t)/T),k=1..m) ;
```

$$SerieFourier := (m, t) \rightarrow \frac{1}{2}\,a_0 + \sum_{k=1}^{m} a_k \cos\left(\frac{2\,k\,\pi\,t}{T}\right) + \sum_{k=1}^{m} b_k \sin\left(\frac{2\,k\,\pi\,t}{T}\right)$$

```
> SerieFourier(5,t):simplify(%);
```

$$\frac{1}{6750}\,\frac{1}{\pi^2}\left(675\,\pi^2+7200\cos\left(\frac{1}{2}\,\pi\,t\right)\sqrt{2}-7200\cos\left(\frac{1}{2}\,\pi\,t\right)-3600\cos(\pi\,t)\right.$$
$$-800\cos\left(\frac{3}{2}\,\pi\,t\right)-800\cos\left(\frac{3}{2}\,\pi\,t\right)\sqrt{2}-1800\cos(2\,\pi\,t)$$
$$\left.-288\cos\left(\frac{5}{2}\,\pi\,t\right)-288\cos\left(\frac{5}{2}\,\pi\,t\right)\sqrt{2}\right)$$

```
> SerieFourier(10,t):simplify(%);
```

$$\frac{1}{2976750}\,\frac{1}{\pi^2}\left(297675\,\pi^2+3175200\cos\left(\frac{1}{2}\,\pi\,t\right)\sqrt{2}-3175200\cos\left(\frac{1}{2}\,\pi\,t\right)\right.$$
$$-1587600\cos(\pi\,t)-352800\cos\left(\frac{3}{2}\,\pi\,t\right)-352800\cos\left(\frac{3}{2}\,\pi\,t\right)\sqrt{2}$$
$$-793800\cos(2\,\pi\,t)-127008\cos\left(\frac{5}{2}\,\pi\,t\right)-127008\cos\left(\frac{5}{2}\,\pi\,t\right)\sqrt{2}$$
$$-176400\cos(3\,\pi\,t)+64800\cos\left(\frac{7}{2}\,\pi\,t\right)\sqrt{2}-64800\cos\left(\frac{7}{2}\,\pi\,t\right)$$
$$\left.+39200\cos\left(\frac{9}{2}\,\pi\,t\right)\sqrt{2}-39200\cos\left(\frac{9}{2}\,\pi\,t\right)-63504\cos(5\,\pi\,t)\right)$$

```
> plot([f,SerieFourier(3,t),SerieFourier(7,t), SerieFourier(10,
  t)],t=-T..T,y=-0.3..0.3,title="Aproximaciones de Fourier: n=3
  (Verde),n=7 (Azul), n=20 (negro) ",color=[red,green,blue,
  black],numpoints=100);
```




```
> SeProm:=evalf(a[0]/2):
> Amp_a0:=plot([[0,0],[0,SeProm]],thickness=3 ):
  Amp_coef:=[seq(plot([[n,0],[n,A[n]]]),n=1..N)]:
  display(Amp_a0,Amp_coef,title=`Espectro de la Señal`);
```



Onda cuadrada y Función Teta de Heaviside

Consideremos la siguiente onda cuadrada

```
> restart;assume(n,integer):with(plots):
  setoptions(thickness=2): # se hacen las lineas mas gruesas
> T :=2; y0 := 1;
```

$$T := 2$$

$$y0 := 1$$

```
> OndaCuad[0] := y0*(Heaviside(t)-Heaviside(t-T/2))
                                  -y0*(Heaviside(t + T/2)-Heaviside(t))
  ;
```

$$OndaCuad_0 := 2\,\mathrm{Heaviside}(t) - \mathrm{Heaviside}(t-1) - \mathrm{Heaviside}(t+1)$$

```
> plot(OndaCuad[0],t=-2..2,y=-2..2,title="La Onda Cuadrada
  Basica");
```




```
> NT :=3:
  for i from -NT to NT do
            tt := t - i*T;
            OndaCuad[i] := y0*(Heaviside(tt)-Heaviside(tt-T/2))
                              -y0*(Heaviside(tt +T/2)-Heaviside(tt));
  end do:
> plot(sum(OndaCuad[j],j=-NT..NT),t=-4..4,y=-2..2,
              title="La Onda Cuadrada de periodo L");
```




```
> f1 := -1;
  f2 := 1;
```

$$f1 := -1$$

$$f2 := 1$$

El coeficiente a_0 y los coeficientes pares a[n] se anulan porque la funci'on es impar

```
> a[0]:=(2/T)*Int('f(t)',t=-T/2..T/2) = (2/T)*(int(f1,t=-T/2..0)+
  int(f2,t=0..T/2));A[0]:=rhs(%);
```

$$a_0 := \int_{-1}^{1} f(t)\, dt = 0$$

$$A_0 := 0$$

```
> a[n]:=(2/T)*Int('f(t)'*cos((2*n*t*Pi)/(T)),t=-T/2..T/2) =
  simplify((2/T)*(int(f1*cos((2*n*t*Pi)/(T)),t=-T/2..0)+int(f2*
  cos((2*n*t*Pi)/(T)),t=0..T/2)));
```

$$a_{n\sim} := \int_{-1}^{1} f(t) \cos(n\sim t \pi)\, dt = 0$$

```
> A[k]:=subs(n=k,simplify((2/T)*(int(f1*cos((2*n*t*Pi)/(T)),t=-
  T/2..0)+int(f2*cos((2*n*t*Pi)/(T)),t=0..T/2))));
```

$$A_k := 0$$

sobreviven los coeficientes impares

```
> b[n]:=(2/T)*Int('f(t)'*sin((2*n*t*Pi)/(T)),t=-T/2..T/2) =
  simplify((2/T)*(int(f1*sin((2*n*t*Pi)/(T)),t=-T/2..0)+int(f2*
  sin((2*n*t*Pi)/(T)),t=0..T/2)));
```

$$b_{n\sim} := \int_{-1}^{1} f(t)\sin(n\sim t\,\pi)\,\mathrm{d}t = \frac{2\left((-1)^{1+n\sim}+1\right)}{n\sim\,\pi}$$

```
> B[k]:=subs(n=k,simplify((2/T)*(int(f1*sin((2*n*t*Pi)/(T)),t=-
  T/2..0)+int(f2*sin((2*n*t*Pi)/(T)),t=0..T/2))));
```

$$B_k := \frac{2\left((-1)^{1+k}+1\right)}{k\,\pi}$$

```
> SerieFourier := (m,t)->
                      A[0]/2 +
                      sum(A[k]*cos((2*k*Pi*t)/T),k=1..m) +
              sum(B[k]*sin((2*k*Pi*t)/T),k=1..m)   ;
```

$$SerieFourier := (m,t) \to \frac{1}{2}A_0 + \sum_{k=1}^{m} A_k \cos\left(\frac{2\,k\,\pi\,t}{T}\right) + \sum_{k=1}^{m} B_k \sin\left(\frac{2\,k\,\pi\,t}{T}\right)$$

```
> SerieFourier(5,t);
```

$$\frac{4\sin(\pi\,t)}{\pi} + \frac{4}{3}\,\frac{\sin(3\,\pi\,t)}{\pi} + \frac{4}{5}\,\frac{\sin(5\,\pi\,t)}{\pi}$$

```
> SerieFourier(10,t);
```

$$\frac{4\sin(\pi\,t)}{\pi} + \frac{4}{3}\,\frac{\sin(3\,\pi\,t)}{\pi} + \frac{4}{5}\,\frac{\sin(5\,\pi\,t)}{\pi} + \frac{4}{7}\,\frac{\sin(7\,\pi\,t)}{\pi} + \frac{4}{9}\,\frac{\sin(9\,\pi\,t)}{\pi}$$

```
> plot([OndaCuad[0],SerieFourier(3,t),SerieFourier(7,t),
  SerieFourier(20,t)],t=-1..1,title="Higher approximations: n=3
  (Green),n=7 (Blue)",color=[red,green,blue,black],numpoints=100)
  ;
```



Series de Fourier y valor absoluto

Muestre y estudie que la expansión de Fouries de

```
> restart; interface(showassumed=0);
```

$$1$$

Definimos |x| y calculamos los coeficientes de Fourier

```
> f:=abs(x);
```

$$f := |x|$$

```
> assume(k,integer);
  c:=Int(f*exp(I*k*x),x=-Pi..Pi)/(2*Pi);
>  c:=int(f*exp(I*k*x),x=-Pi..Pi)/(2*Pi);
```

$$c := \frac{1}{2}\,\frac{\int_{-\pi}^{\pi} |x|\, e^{I\,k\sim\,x}\,dx}{\pi}$$

$$c := -\frac{1}{2}\,\frac{2\,(-1)^{1+k\sim}+2}{k\sim^2\,\pi}$$

Simplificando:

```
> c:=simplify(c);
```

$$c := \frac{(-1)^{k\sim} - 1}{k\sim^2 \pi}$$

Por lo tanto la serie de Fourier será

```
> F:=(n,x)-> Pi/2 - 4*add(cos((2*k-1)*x)/((2*k-1)^2),k=1..n)/Pi;
```

$$F := (n, x) \to \frac{1}{2}\pi - \frac{4\, add\left(\frac{\cos((2k-1)x)}{(2k-1)^2}, k = 1..n\right)}{\pi}$$

donde $\frac{\text{Pi}}{2}$ para el promedio de of |x|.

Graficando

```
> plot([f,F(1,x),F(2,x),F(5,x)],x=-Pi..Pi,title="Primera, 2da y 5ta Suma Parcial",color=[black,red,blue,green]);
```

Primera, 2da y 5ta Suma Parcial




```
> plot([f,F(6,x),F(10,x),F(20,x)],x=0..Pi,title="6ta, 10ma, 20ma Suma Parcial",color=[black,red,blue,green]);
```



De finimos la función $E(n)$ como la resta de las sumas parciales F(n,x) y el valor exacto de la función para $n = 1$ y 2

```
> E:=proc(n) local g;Int( ( f-F(n,x))^2,x=-Pi..Pi);
> g:=F(n,x); g:=expand((abs(x)-g)^2);
> evalf(Int(g,x=-Pi..Pi)); end;
```

$E := \textbf{proc}(n)$

$\quad \textbf{local } g;$

$\quad Int(\ (f - F(n,x))\text{^}2, x = -\text{Pi}..\text{Pi});$

$\quad g := F(n,x);$

$\quad g := expand(\ (\text{abs}(x) - g)\text{^}2);$

$\quad evalf(Int(g, x = -\text{Pi}..\text{Pi}))$

end proc

Para $n = 1$, 2 y 5

```
> E(1); E(2); E(5);
```

$$0.07475460111$$

$$0.01187857421$$

$$0.0008324117948$$

Si calculamos $\log(E_n)$

y mirando la solución anterior intuimos que E_n decrece como n^{-3} si graficamos $n^3\ E_n$ debería ser una constante

```
> es:=NULL: for k from 1 to 9 do;
> z:=E(k); es:=es,[k,log[10](z)];
> printf(`For N=%3.0f  el error es %11.8f\n`,k,z*k^3); od:es:=
  [es]:
For N=  1  el error es  0.07475460
For N=  2  el error es  0.09502859
For N=  3  el error es  0.10070571
For N=  4  el error es  0.10295418
For N=  5  el error es  0.10405147
For N=  6  el error es  0.10466408
For N=  7  el error es  0.10503937
For N=  8  el error es

Error, (in fprintf) number expected for floating point format
```

La gráfica del logaritmo

```
> plot(es);
```



Finalmente graficamos |x| para varias sumas parciales

```
> xmax:=2*Pi:  plot([f,F(1,x)],x=-xmax..xmax,view=[-xmax..xmax,0.
```

```
.5],numpoints=200,title="1era suma parcial",colour=[red,blue]);
```




```
> xmax:=2*Pi:  plot([f,F(2,x)],x=-xmax..xmax,view=[-xmax..xmax,0..5],numpoints=200,title="2da suma parcial",colour=[red,blue]);
```




```
> xmax:=2*Pi:  plot([f,F(4,x)],x=-xmax..xmax,view=[-xmax..xmax,0.
.5],numpoints=200,title="4ta suma parcial",colour=[red,blue]);
```




```
> xmax:=2*Pi:  plot([f,F(6,x)],x=-xmax..xmax,view=[-xmax..xmax,0..5],numpoints=200,title="6ta suma parcial",colour=[red,blue]);
```




```
> xmax:=2*Pi:  plot([f,F(10,x)],x=-xmax..xmax,view=[-xmax..xmax,
  0..5],numpoints=200, title="10ma suma parcial",colour=[red,
  blue]);
```



Cerca del origen

```
> xmax:=0.2:  plot([f,F(10,x),F(50,x)],x=0..xmax,view=[0..xmax,0.
.xmax],numpoints=200,title="10ma(roja) y 50(azul) suma
parcial",color=[black,red,blue]);
```



Consideremos la diferencia entre la décima y la 50 suma parcial para $f = |x|$.

```
> plot([f-F(10,x),f-F(50,x)],x=0..Pi,color=[red,blue]);
```



Función contínua a trozos

Hacemos lo que hicimos anteriomente para otro ejemplo

```
> restart;
> f:=x->piecewise( x<= 0, x/Pi, 1-x/Pi);
```

$$f := x \rightarrow piecewise\left(x \leq 0, \frac{x}{\pi}, 1 - \frac{x}{\pi}\right)$$

```
> plot(f(x),x=-Pi..Pi);
```



Encontramos los coeficientes de fourier y vemos sus partes reales e imaginarias

```
> assume(k,integer);
> c:=(int(f(x)*exp(I*k*x),x=-Pi..Pi)/(2*Pi));
```

$$c := \frac{1}{2}\,\frac{-\frac{(-1)^{k\sim} + \mathrm{I}\,(-1)^{k\sim}\,\pi\,k\sim - 1}{k\sim^2\,\pi} + \frac{1 + \mathrm{I}\,\pi\,k\sim - (-1)^{k\sim}}{k\sim^2\,\pi}}{\pi}$$

```
> simplify(evalc(Re(c)));
```

$$-\frac{(-1)^{k\sim} - 1}{\pi^2\,k\sim^2}$$

```
> simplify(evalc(Im(c)));
```

$$\frac{1}{2}\,\frac{(-1)^{1 + k\sim} + 1}{\pi\,k\sim}$$

el valor medio de la función

```
> co:=int(f(x),x=-Pi..Pi)/(2*Pi);
```

$$co := 0$$

La serie de Fourier será

```
> F:=(n,x)-> 4*add(cos((2*k-1)*x)/((2*k-1)^2), k=1..n)/Pi^2 + 2*
```

```
add(sin((2*k-1)*x)/(2*k-1),k=1..n)/Pi;
```

$$F := (n, x) \to \frac{4\, add\left(\frac{\cos((2\,k-1)\,x)}{(2\,k-1)^2}, k = 1 \,..n\right)}{\pi^2} + \frac{2\, add\left(\frac{\sin((2\,k-1)\,x)}{2\,k-1}, k = 1 \,..n\right)}{\pi}$$

la 100 suma parcial en $x = 0$, $x = \text{Pi}$ y $x = -\text{Pi}$

```
> evalf(F(100,0)); evalf(F(100,Pi)); evalf(F(100,-Pi));
```

$$0.4989867964$$

$$-0.4989867964$$

$$-0.4989867964$$

comparamos la periodicidad de las sumas parciales

```
> xmax:=2*Pi:plot([f(x),F(20,x)],x=-xmax..xmax,view=[-xmax..xmax,
  -1.5..1.5],color=[black,red]);
```



Sumamos las sumas parciales alrededor de $x = 0$

```
> plot([f(x),F(40,x),F(100,x),F(200,x)],x=-0.1..0.1,colour=
```

```
[black,red,green,blue]);
```